

# **Pspice v9.1 student edition**

**Introduzione all'uso**



## Cosa è

- PSpice (Simulation Program with Integrated Circuit Emphasis) è un programma di uso generale per la simulazione di circuiti elettrici.
- Nato nel 1984 dallo Microsim appartiene alla famiglia di versioni commerciali di SPICE

# Unità di misura in gioco

Nome	Simbolo	Valore
Tera	T	1e12
Giga	G	1e9
Mega	Meg	1e6
Kilo	K	1e3

Nome	Simbolo	Valore
Milli	m	1e-3
Micro	u	1e-6
Nano	n	1e-9
Pico	p	1e-12
Feta	f	1e-15

## Unità di misura in gioco 2

- Quando si costruiranno i circuiti elettrici verranno inseriti i simboli adeguati per le unità di misura.
- Es.
  - Un resistore da 2.2 KOhm viene scritto
    - 2200
    - 2.2K
    - 2.2e3
    - 22e2
  - Pspice non considera le lettere che non conosce. Esempio: 2.2j viene considerato 2.2 Ohm

## Struttura dei file di ingresso

- Pspice lavora con file di questo genere:
  - Nome\_file.cir
  - Nome\_file.dat
- Nome\_file.cir ha al suo interno la descrizione del circuito
- Nome\_file.dat contiene le informazione relative alla simulazione (viene generato automaticamente da pspice)

## Struttura dei file di ingresso 2

- Dal programma vengono generati anche altri file, che posso contenere diverse informazioni.
  - Esempio:
    - gli errori che sono stati commessi nella scrittura del file.

## Struttura dei file di ingresso 3

- Il file: **nome\_file.cir** viene scritto come un normale file di testo.
- La prima riga **NON** deve contenere comandi ma il titolo o una qualunque informazione
- Il resto del listato si suddivide in:
  - Linee elemento (componenti)
  - Linee comando (Tipo di analisi, parametri...)
- Le linee elemento e le linee comando non sono case-sensitive
  - Es.
    - V1 si può anche scrivere v1
    - H12bys coincide con H12Bys, h12BYS etc.
- L'ultima riga deve contenere il comando:
  - **.END**

# Struttura dei file di ingresso 4


- Il circuito si costruisce per nodi.
- L'ordine con cui si scrivono i componenti del circuito è indifferente
- Per le correnti e le tensioni di ramo, si utilizza la convenzione degli utilizzatori:
  - Fissati il nodo + e –
  - Il verso positivo della corrente è quello dal nodo + al nodo -, attraverso il dispositivo.
- Ogni nodo deve essere numerato in modo sequenziale senza salti di numerazione.
- Il nodo 0 viene considerato il nodo di riferimento a massa.
- Ogni nodo deve avere almeno due elementi ad esso connesso.

# Presentazione di PSPICE

Quando si effettua la stesura di un circuito, prima di far partire la simulazione bisogna salvare il file: Nomefile.cir . Il passo successivo è caricare il listato salvato e far partire la simulazione:

Menu:


- Simulation
- RUN



# Presentazione di PSPICE – 2

Una volta fatta partire la simulazione viene aperto un secondo programma, PROBE. Da questo per aggiungere le forme d'onda basta cliccare:

- Trace
    - Add trace
- E cliccare sui nodi (tensioni, o correnti) per visualizzare i risultati.



# Capitolo 1

- Contenuti in questo capitolo
  - Le linee elemento
    - Resistenze
    - Capacità
    - Induttori
    - Generatori indipendenti
    - Generatori dipendenti
    - Possibilità offerte dai generatori indipendenti
    - Stesura di sottocircuiti
  - Le linee comando
    - Analisi in frequenza
    - Analisi in continua
    - Analisi in transitorio
    - Stampare i risultati
    - Terminare il listato
  - Esempi circuitali

# Linee elemento 1

- Resistori:
  - Forma generale:
    - Rxxxxxxx N+ N- Valore <Tc=Tc1<,TC2>>
  - Xxxxxxx identificativo resistore
  - N+ N- connessione ai nodi
  - Valore: il valore del resistore
  - <Tc=Tc1<,TC2>>: non illustrato in queste slide
- Esempi:
  - R2 1 3 10K
  - RC1 12 17 1K

# Linee elemento 2

- Condensatori
  - Cxxxxxx N+ N- Valore <IC = INCOND>
    - <IC = INCOND> sono le condizioni iniziali espresse in volt.
  - Esempi:
    - C1 2 3 10uF
    - Csc 14 12 1nF IC=3V
    - Cbya 21 34 1m IC = 2
- Induttori:
  - Lxxxxxx N+ N- Valore <IC = INCOND>
    - <IC = INCOND> sono le condizioni iniziali espresse in ampere. Se non inserite vengono automaticamente messe a 0 (induttore scarico)
  - Esempi:
    - L1 4 10 1mH
    - L3 2 1 1n
    - Lbys 4 1 1m IC = 2mA

# Linee elemento 3

- **Generatori di tensione:**
  - Vxxxxxx N+ N- <DC Valore> <Valore in transitorio> + <AC <ACMAG <ACPHASE>>>
    - <DC Valore> Contiene il valore del generatore di tensione
      - Come si può vedere questo valore è opzionale. Se non inserito viene assunto 0.
      - La sigla DC può essere omessa.
      - <Valore in transitorio> descrive la forma d'onda da assumersi in transitorio
      - <AC ...> indica la simulazione in regime sinusoidale
        - ACMAG ampiezza di analisi
        - ACPHASE fase in analisi
- **Generatori di corrente:**
  - Ixxxxxx N+ N- <DC Valore> <Valore in transitorio> + <AC <ACMAG <ACPHASE>>>
- **Esempi:**
  - Vcc 1 0 DC 6
  - Vin 13 2 0.001 AC 1 SIN(0 1 1MEG)
  - I 23 41 5

## Linee elemento 4

- Generatore di tensione dipendenti:
  - Gen tensione controllato in tensione
    - Forma generale:
      - Exxxxxxx N+ N- NC+ NC- Valore
        - N+ N- sono i nodi di connessione positivo e negativo
        - NC+ NC- sono i nodi ai quali si misura la tensione che controlla il generatore
        - Valore è l'amplificazione di tensione
    - Esempi:
      - E1 1 2 4 6 1E18

## Linee elemento 5

- Generatore di tensione dipendenti:
  - Gen tensione controllato in corrente:
    - Forma generale:
      - Hxxxxxxx N+ N- VNAM Valore
        - VNAM è il nome di un generatore di tensione attraverso cui scorre la corrente di controllo.
        - Valore contiene la transresistenza in ohm
    - Esempi:
      - Hx 5 12 Vz 0.2K

## Linee elemento 6

- **Generatori di corrente dipendenti:**
  - Gen corrente controllato in corrente:
    - Forma generale:
      - Fxxxxxx N+ N- VNAM Valore
        - Valore è l'amplificazione di corrente
  - Gen corrente controllato in tensione:
    - Forma generale:
      - Gxxxxxx N+ N- NC+ NC- Valore
        - Valore è la transconduttanza in siemens.

## Possibilità offerte dai generatori indipendenti

- Impulso
  - PULSE(V1 V2 TD TR TF PW PER)
    - V1 tensione minima [volt o ampere]
    - V2 tensione massima [volt o ampere]
    - TD tempo di ritardo [secondi]
    - TR tempo di salita [secondi]
    - TF tempo di discesa [secondi]
    - PW durata dell'impulso allo stato V2 [secondi]
    - PER periodo dell'impulso [secondi]
  - Esempi:
    - Vin 1 0 PULSE(-1 1 2ns 2ns 50ns 100ns)

## Possibilità offerte dai generatori indipendenti 2

- Sinusoide
  - `SIN(V0 VA FREQ TD THETA PHASE)`
    - V0 valore di offset [volt o ampere]
    - VA valore di picco [volt o ampere]
    - TD tempo di ritardo [secondi]
    - THETA coefficiente di smorzamento [1 / secondi]
    - PHASE fase [gradi]
  - Esempi:
    - `Vin 1 0 SIN(5 1m 1meg 0 0 0)`
    - `V2 4 5 SIN(0 10 1k 0 1e10 90)`
  - Il fattore di smorzamento THETA segue la legge:  
 $VA \exp[-(t-TD)THETA]$

## Possibilità offerte dai generatori indipendenti 3

- **Esponenziale**
  - EXP(V1 V2 TD1 TAU1 TD2 TAU2)
    - V1 valore minima
    - V2 valore massimo
    - TD1 tempo di ritardo in salita
    - TAU1 costante di tempo in salita
    - TD2 tempo di ritardo in discesa
    - TAU2 costante di tempo in discesa
  - Esempi:
    - Vin 1 0 exp(-4 -1 2ns 30ns 60ns 40ns)

## Possibilità offerte dai generatori indipendenti 4

- **Lineare a tratti**
  - PWL(T1 V1 T2 V2 T3 V3 T4 V4 ... Tn Vn)
    - Ogni coppia di valore  $T_i$   $V_i$  rappresenta il valore che assume tensione o corrente (a seconda di che generatore viene utilizzato) all'istante  $T_i$
- **Esempi:**
  - $V_{in}$  1 0 PWL(0 -7 1n -5 2n 1 5n -5 6n -3 7n 0 8n 5)

## Possibilità offerte dai generatori indipendenti 5

- Frequenza singola FM
  - SFFM( $V_0$  VA FC MDI FS)
    - $V_0$  valore di offset
    - VA ampiezza
    - FC frequenza portante
    - MDI indice di modulazione
    - FS frequenza del segnale
  - Esempi:
    - Vin 1 0 SFFM(0 1M 20K 5 1K)
  - La forma d'onda è data dall'espressione
$$V(t) = V_0 + VA \sin[(2\pi FC t) + MDI \sin(2\pi FS t)]$$

## Sottocircuiti – Sub Circuit

---

- In pspice è possibile creare dei sottocircuiti.
- Un sottocircuito è un insieme di componenti che formano un dispositivo.
- Questo può essere usato nello stesso file in cui è stato creato oppure importato in un altro file.

# Sottocircuiti – Sub Circuit 2

- Forma generale di richiamo sottocircuito
  - Xyyyyyy N1 <N2 N3 ... Nn> SUBNAME
    - Esempi:
      - X1 2 3 5 6 4 opamp
      - Xbys 12 43 32 sub1
- Definizio di un sottocircuito
  - .SUBCKT SUBNAME N1 <N2 N3 ... Nn>
    - Contenuto del sottocircuito
  - .ENDS SUBNAME
    - Esempi:
      - .SUBCKT opamp 1 2 3
      - E1 3 0 1 2 1E18
      - .END opamp

## Linee comando 1

- La linea `.TRAN` richiede un'analisi in transitorio del circuito in esame. Il tempo di default parte dall'istante `t=0` e termina a piacere dell'utente.
- `*.TRAN TSTEP TSTOP <TSTART <TMAX>> <UIC>`
  - `TSTEP` è l'intervallo di tempo tra un punto e il successivo della risposta che si sta calcolando
  - `TSTOP` è l'istante di terminazione `t=0 t= TSTOP`
  - `TSTART` è l'istante di partenza della simulazione `t=TSTART t = TSTOP`
  - `TMAX` è il massimo passo di integrazione che verrà utilizzato durante l'integrazione numerica, in caso di sua assenza `pspice` sceglie il più conveniente.

## Linee comando 2

- La linea `.DC` chiede una simulazione in continua con generatori che variano a tratti
- Forma generale
  - `.DC SRCNAM VSTART VSTOP VINCRC <SCR2 START2 ...>`
  - SRCNAM nome del generatore da controllare
  - VSTART valore iniziale del generatore
  - VSTOP valore finale del generatore
  - VINCRC passo di incremento tra VSTART VSTOP
- Esempi:
  - `.DC VCE 0 5 100m`
  - `.DC VDS 0 10 .5 VGS 0 5 1`

## Linee comando 3

- La linea .AC richiede un'analisi per piccoli segnali in regime sinusoidale in un campo di frequenze specificato dall'utente. In pratica si effettua la richiesta di un risposta in frequenza.
- Forma generale:
  - .AC DEC ND FSTART FSTOP
  - .AC OCT NO FSTART FSTOP
  - .AC LIN NP FSTART FSTOP
- FSTART frequenza di inizio
- FSTOP frequenza di fine
- DEC per decadi
  - ND punti per decade
- OCT per ottave
  - NO punti per ottave
- LIN in modo lineare
  - NP punti

## Linee comando 4

- \* indica un commento
- .PROBE <Variabili di uscita>
  - Effettua la stampa del grafico della simulazione
- .END
  - Da inserire alla fine del listato del circuito

# Primo circuito – Carica condensatore

- Listato:

Carica di un condensatore

```
V1 1 0 5Vdc
```


```
R1 1 2 1K
```

```
C1 2 0 1n IC = 0
```


```
.TRAN 1N 6u
```

```
.PROBE V(2)
```

```
.END
```



# Primo circuito - PROBE



Tensione sul condensatore

## Secondo circuito: carica di un induttore

- Listato:

Carica di un induttore

```
V1 1 0 5V
```


```
R1 1 2 1k
```

```
L1 2 0 1m IC=0
```


```
.TRAN 1n 6u
```

```
.PROBE
```

```
.END
```



# Secondo circuito - PROBE




Corrente nell'induttore


## Terzo circuito: Scarica elettrostatica

- Listato:

```
Scarica elettrostatica  
R1 0 1 300  
C1 1 2 50pF IC=20K  
L1 2 0 0.1uH IC=10m  
.TRAN 0.1n 60n  
.PROBE  
.END
```





# Terzo circuito: Scarica elettrostatica - PROBE




## Quarto Circuito: Amplificatore Operazionale

- Listato:
  - Amplificatore operazionale ideale
  - V1 1 0 SIN(0 1mV 1KHz)
  - \*Generatore di onda sinusoidale 1mv di picco
  - \*senza offset ampiezza 1m e frequenza 1Khz
  - X1 1 2 3 OPAMP
  - \* + - Out ----- Ingressi / Uscite Opamp
  - R2 3 2 10K
  - R1 2 0 1K
  - .SUBCKT OPAMP 1 2 3**
  - \*----- + INGRESSO NON INVERTENTE**
  - \*----- - INGRESSO INVERTENTE**
  - \*----- 0 USCITA**
  - E1 3 0 1 2 1E8**
  - .ENDS OPAMP**
  - .tran 1u 1m
  - .probe
  - .end



## Quarto Circuito: Amplificatore Operazionale - PROBE



# Capitolo 2

- **Assegnare valore alle variabili**
  - Sintassi generale
  - Variare le variabili automaticamente
- **Specificare un modello di componente**
  - La linea .model
  - **Dispositivi a semiconduttore**
    - **Diodi**
      - Parametri possibili
    - **Bipolari**
      - Parametri possibili
    - **MOSFET**
      - Parametri possibili
    - **JFET**
      - Parametri possibili
  - **Dispositivi generici**
    - Interruttore controllato in tensione
    - Interruttore controllato in corrente
- **Esempi**

# Assegnare valori a variabili

- **Forma generale:**
  - .PARAM name1 = val1 <, val2, ...>
  - .PARAM name1 = {espressione} <, nome2 = {espressione2}, >
- **Esempio:**
  - .SUBCKT FILTRO IN OUT PARAMS: FO=300KHZ BW = 150
    - La parola chiave PARAMS: pone termine alla lista dei nodi e assegna alle successive variabili FO e BW i rispettivi valori.
    - La successiva chiamata al sottocircuito può contenere valori diversi:
      - X1 1 5 FILTRO PARAMS: BW=300

# Variare le variabili

- **Forma generale:**
  - `.STEP VARNAM LIST val1 val2 val3 ...`
  - `.STEP <LIN> VARNAM val1 va2 incr`
  - `.STEP <DEC> VARNAM val1 val2 np`
  - `.STEP <OCT> VARNAM val1 val2 np`
- Il comando permette di far variare la variabile VARNAM in un campo di valori assegnato.
- **Esempio:**
  - `.STEP VCE 0 10V 1V`
  - `.STEP LIN I2 -0.5mA 6mA 0.5mA`
- Questo esempio simula un resistore che varia da 30 a 50 Ohm a passi di 5.
- `.PARAM RVAL = 1`
- `R1 1 2 {RVAL}`
- `.STEP PARAM RVAL 30 50 5`

## Specificare un modello di componente

- La linea `.MODEL` specifica un insieme di valori per i parametri di un modello o di un dispositivo.
- Forma generale:
  - `.MODEL MNAME TYPE(PNAME=VAL1 ...)`
- Esempio:
  - `.MODEL MOD1 NPN BF=50 IS =1e-13 VBF=50`

## Specificare un modello di componente 2

- I tipi possibili per i modelli dei dispositivi a semiconduttore usati in PSPICE:

Nome	Dispositivo
D	Diodo
NPN	Transistor NPN
PNP	Transistor PNP
NJF	N-Channel JFET
PJF	P-Channel JFET
NMOS	N-Channel MOSFET
PMOS	P-Channel MOSFET

# Parametri per il diodo

Nome	Parametro	Unità	Default	Esempio
IS	corrente di saturazione	A	10-14	10-15
RS	resistenza parassita		0	10
N	coefficiente di emissione		1	1
TT	tempo di transito	s	0	0.1
CJO	capacità $pn$ senza polarizzazione	F	0	2 p
VJ	potenziale di giunzione $pn$	V	1	0.6
M	coefficiente di gradualità della giunzione	$pn$	0.5	0.5
EG	ampiezza della banda proibita	eV	1.11	1.11
XTI	Coefficiente di temperatura di $I_S$		3 3	
KF	coefficiente del rumore flicker		0	
AF	esponente del rumore flicker		1	
FC	coefficiente della capacità di svuotamento		0.5	
BV	tensione inversa di breakdown	V	50	
IBV	corrente inversa di breakdown	A	1E-10	

# Parametri per il bjt

Nome	Parametro	Unità	Default	Esempio
IS	Corrente di saturazione	A	10–16	10–16
NF	Coefficiente di emissione diretto		1	1
BF	Valore massimo ideale del guadagno diretto		100	100
IKF	Valore di spigolo per la riduzione del guadagno diretto	A	$\infty$	10M
NE	Coefficiente di emissione di perdita	BE	1.5	2
VAF	Tensione di Early diretta	V	$\infty$	100
BR	Valore massimo ideale del guadagno inverso		1	0.1
NR	Coefficiente di emissione inverso		1	1
IKR	Valore di spigolo per la riduzione del guadagno inverso	A	$\infty$	100M
RB	Resistenza ohmica di base	$\Omega$	0	100
RC	Resistenza ohmica di collettore	$\Omega$	0	10
RE	Resistenza ohmica di emettitore	$\Omega$	0	1
TF	Tempo di transito diretto	s	0	0
TR	Tempo di transito inverso	s	0	0
CJE	Capacità BE senza polarizzazione	F	0	2p

# Parametri per il bjt - 2

Nome	Parametro	Unità	Default	Esempio
VJE	Potenziale della giunzione BE	V	0.75	0.7
MJE	Coefficiente di gradualità della giunzione BE		0.33	0.33
FC	Coefficiente per la capacità di svuotamento in polarizzazione diretta		0.5	
CJC	Capacità BC senza polarizzazione	F	0	1p
VJC	Potenziale della giunzione BC	V	0.75	0.5
MJC	Coefficiente di gradualità della giunzione BC		0.33	0.33
CJS	Capacità senza polarizzazione della giunzione C-substrato	F	0	2p
VJS	Potenziale della giunzione C-substrato	V	0.75	0.75
MJS	Coefficiente di gradualità della giunzione C-substrato		0	0
EG	Ampiezza della banda proibita	eV	1.11	1.11
XTI	Coefficiente di temperatura della corrente di saturazione		3	
KF	Coefficiente del rumore flicker		0	
AF	Esponente del rumore flicker		1	1

# Parametri per il JFET

Nome	Parametro	Unità	Default	Esempio
VTO	Tensione di soglia	V	-2	-2
BETA	Parametro della transconduttanza	A/V <sup>2</sup>	1E-4	1E-3
LAMBDA	Fattore di modulazione della lunghezza di canale	V-1	0	1E-4
RD	Resistenza ohmica di drain	$\Omega$	0	100
RS	Resistenza ohmica di source	$\Omega$	0	100
IS	Corrente di saturazione di gate <i>pn</i>	A	1E-14	1E-14

# Parametri per il JFET - 2

Nome	Parametro	Unità	Default	Esempio
PB	Potenziale di gate $p_n$	V	1	0.6
CGD	Capacità GD senza polarizzazione	F	0	5p
CGS	Capacità GS senza polarizzazione	F	0	1p
FC	Coefficiente per la capacità di svuotamento in polarizzazione diretta		0.5	
KF	Coefficiente del rumore flicker		0	
AF	Esponente del rumore flicker		1	

# Parametri per il MOSFET

Nome	Parametro	Unità	Default	Esempio
LEVEL	Indice del modello (default: Shichman-Hodges)		1	
L	Lunghezza di canale	m	DEFL	
W	Larghezza di canale	m	DEFW	
VTO	Tensione di soglia	V	0	0.1
KP	Parametro della transconduttanza	A/V <sup>2</sup>	2E-5	2.5E-5
LAMBDA	Fattore di modulazione della lunghezza di canale	V-1	0	0.02
GAMMA	Parametro della soglia del substrato	V <sup>1/2</sup>	0	0.35
RD	Resistenza ohmica di drain	$\Omega$	0	10
RS	Resistenza ohmica di source	$\Omega$	0	10
IS	Corrente di saturazione di substrato <i>pn</i>	A	1E-14	1E-15
JS	Densità di corrente di saturazione di substrato <i>pn</i>	A/m <sup>2</sup>	0	1E-8
PB	Potenziale di substrato <i>pn</i>	V	0.8	0.75
CBD	Capacità D-substrato senza polarizzazione	F	0	5p
CBS	Capacità S-substrato senza polarizzazione	F	0	2p
CJ	Capacità per unità di area del substrato	F/m <sup>2</sup>	0	
MJ	Coefficiente di gradualità della giunzione di substrato		0.5	

# Parametri per il MOSFET – 2

Nome	Parametro	Unità	Default	Esempio
FC	Coefficiente per la capacità di svuotamento in polarizzazione diretta		0.5	
CGS0	Capacità G-S senza polarizzazione per unità di lunghezza	F/m	0	
CGD0	Capacità G-D senza polarizzazione per unità di lunghezza	F/m	0	
CGB0	Capacità G-substrato senza polarizzazione per unità di lunghezza	F/m	0	
TOX	Spessore dell'ossido	m	10–5	10–6
NSUB	Drogaggio del substrato	unità/cm3	0	0
UO	Mobilità superficiale	cm2/Vs	600	600
KF	Coefficiente del rumore flicker		0	1E-26
AF	Esponente del rumore flicker		1	1.2

# Componenti generici

- Interruttore controllato in tensione
  - Forma generale:
    - Sxxxxxx N+ N- NC+ NC- MNAME
    - .MODEL MNAME VSWITCH <PNAME1=PVAL1 PNAME2=PVAL2 ... >
  - Esempi:
    - S1 2 3 5 0 SMOD
    - .MODEL SMOD VSWITCH ROFF=1E10
  - L'interruttore viene visto come una resistenza di due valori RON ROFF. Quando la tensione di controllo NC+ NC- raggiunge una certa soglia l'interruttore si chiude in RON (solitamente molto bassa) in caso contrario si apre (solitamente molto alta).
  - Se RON ROFF VON VOFF non vengono indicati, sono utilizzati i valori di default:
    - RON = 1 ohm
    - ROFF = 1 Megohm
    - Von = 1 Volt
    - Voff = 0 Volt

## Componenti generici 2

- Interruttore controllato in corrente
  - Forma generale:
    - Wxxxxxx N+ N- VNAM MNAME
    - .MODEL MNAME ISWITCH <PNAME1=PVAL1 ... >
  - VNAM è il generatore di tensione in cui scorre la corrente di controllo.
  - In assenza di RON ROFF ION IOFF vengono assunti i valori di default:
    - RON = 1 Ohm
    - ROFF = 1 MEGOhm
    - ION = 1mAmpere
    - IOFF = 0 Ampere

# Dispositivi a semiconduttore

- I dispositivi utilizzati
  - Diodo a giunzione
  - Transistore a giunzione (BJT)
  - Transistore ad effetto di campo
    - JFET
    - MOSFET

## Dispositivi a semiconduttore 2

- Diodo a giunzione
  - Forma generale:
    - Dxxxxxxx N+ N- MNAME <AREA>
  - Esempio:
    - DBRIDGE 2 10 DIODE1
    - DCLMP 3 7 DMOD 3.0
- N+ è il nodo a cui è connesso l'anodo
- N- è il nodo a cui è connesso il catodo
- MNAME è il nome del modello
- <AREA> è il fattore di area

## Dispositivi a semiconduttore 3

- Transistore a giunzione:
  - Forma generale:
    - Qxxxxxxx NC NB NE <NS> MNAME <AREA>
  - Esempio:
    - Q1 1 2 3 QMOD
  - NC è il nodo a cui è connesso il catodo
  - NB è il nodo a cui è connessa la base
  - NE il nodo a cui è connesso l'emettitore
  - <NS> è il nodo opzionale del substrato

## Dispositivi a semiconduttore 4


- Transistore ad effetto di campo JFET
  - Forma generale:
    - Jxxxxxxx ND NG NS MNAME <AREA>
  - Esempio:
    - J1 8 2 5 JM1
  - ND è il nodo a cui è connesso il drain
  - NG è il nodo a cui è connesso il gate
  - NS è il nodo a cui è connesso il source

# Dispositivi a semiconduttore 5


- MOSFET:
  - Forma generale:
    - Mxxxxxx ND NG NS NB MNAME <L=VAL> <W=VAL> <AD=VAL> +<AS = VAL> <PD=VAL> <PS=VAL> <NRD= VAL> <NRS = VAL>
  - Esempi:
    - M1 23 2 0 20 TYPE1
    - M32 2 18 12 19 MODM L=5U W = 2U
  - ND è il nodo a cui è connesso il drain
  - NG è il nodo a cui è connesso il gate
  - NS è il nodo a cui è connesso il source
  - NB è il nodo a cui è connesso il bulk

# Carica e scarica di un condensatore

```
Carica e scarica condensatore
V1 1 0 5Vdc
S1 1 2 5 0 SMOD
.MODEL SMOD VSWITCH VON = 0 VOFF = 4V
*Generatore per il comando di S1:
V2 5 0 PULSE(0 5 0 1N 1N 6M 12M)
*
R1 2 3 1K
C1 3 0 1U IC=0
S2 3 4 5 0 SMOD2
.MODEL SMOD2 VSWITCH VON = 4 VOFF = 0
R2 4 0 1K
.TRAN 0.01U 36M
.PROBE
.END
```



# Carica e scarica di un condensatore - PROBE



# Amplificatore con guadagno variabile

Amplificatore con guadagno variabile

```
V1 1 0 AC 1 SIN(0 1m 1K)
```

```
R1 1 2 1K
```

\*Definizione resistenza variabile

```
.PARAM RVAL = 1
```

```
R2 2 3 {RVAL}
```


```
.STEP PARAM RVAL 1K 10K 1K
```

```
E1 3 0 1E8
```


```
.TRAN 0.1M 1M
```

```
.PROBE
```

```
.END
```



# Amplificatore con guadagno variabile -PROBE



# Rilevazione caratteristica di un BJT

Rilevazione caratteristiche di un BJT

```
IBE 0 2
```

```
VCE 1 0
```


```
Q1 1 2 0 MOD1
```

```
.MODEL MOD1 NPN(VAF=50 VAR=50)
```


```
.DC VCE 0 10 0.01 IBE 0 200U 100U
```

```
.PROBE
```

```
.END
```




# Rilevazione caratteristica di un BJT - PROBE




# Stadio amplificatore CE


```
Stadio amplificatore CE
Vsignal 1 0 AC 1 SIN(0 1M 1K)
Val 5 0 12Vdc
C1 1 2 10u
R2 2 5 9K
R1 2 0 1K
Q2 3 2 4 MOD1
.MODEL MOD1 NPN
Rc 3 5 10K
Re 4 0 1K
C2 3 6 10u
CL 3 6 10u
RL 0 6 100K
.TRAN 1N 1M
.AC DEC 100 1 100K
.PROBE
.END
```



# Stadio amplificatore CE - PROBE



# Stadio amplificatore CE – AC PROBE



Questa simulazione viene presentata in presenza della linea .AC

# Convertitore AC/DC

Convertitore AC/DC a valor medio convenzionale

```
Vac 1 0 SIN(0 12 50)
```

\*Creazione ponte di diodi

```
D1 2 1 DBRIDGE1
```

```
.MODEL DBRIDGE1 D
```

```
D2 2 0 DBRIDGE2
```

```
.MODEL DBRIDGE2 D
```

```
D3 1 3 DBRIDGE3
```

```
.MODEL DBRIDGE3 D
```

```
D4 0 3 DBRIDGE4
```

```
.MODEL DBRIDGE4 D
```

\*


```
C1 3 2 100U
```

```
C2 3 2 100N
```


```
.TRAN 0.1M 20M
```

```
.PROBE
```

```
.END
```




# Convertitore AC/DC - PROBE




# Stadio di uscita in classe B

```
Stadio di uscita in classe B
Vsignal 1 0 SIN(0 4V 1K)
VDD 2 0 DC 15
VSS 3 0 DC -15
M1 2 1 4 4 N1 W=1000U L=2U
M2 3 1 4 4 P1 W=2500U L=2U
RL 4 0 2K
.MODEL N1 NMOS VTO=1 KP=25E-6
.MODEL P1 PMOS VTO=-1 KP=10E-6
.TRAN 1U 2M
.PROBE
.END
```



# Stadio di uscita in classe B - PROBE



## Capitolo 3

- Librerie
  - Include in PSpice Student Edition v.9.1
  - Includere una libreria
- Esempi
- Ricercare un componente
- Circuiti analogico-numeric
  - Definizione di analogico-numeric
  - Nodi globali e locali
- Esempi

## Librerie incluse in Pspice

- Pspice include delle librerie conteneti sottocircuiti.
- Questi contengono la realizzazione di circuiti realmente esistenti in commercio (ua741, ne555, 2n2222, 1n4148 etc.)
- La versione per studenti include solo una libreria:
  - Eval.lib
- La versione commerciale include una ricca collezione di librerie.
- E' possibile creare delle proprie librerie.

## Librerie incluse in Pspice 2

- L'inclusione di una libreria in Pspice si effettua tramite:
  - .LIB <nomefile>
- Esempi:
  - .LIB
  - .LIB c:\psp\eval.lib

# Librerie incluse in Pspice 3

- Cosa vi è in eval.lib:
  - Sono presenti le dichiarazioni ed i circuiti dei vari componenti
  - Esempio:
    - `.model Q2N2222 NPN(Is=14.34f Xti=3 Eg=1.11 Vaf=74.03 Bf=255.9 Ne=1.307`  
`+Ise=14.34f Ikf=.2847 Xtb=1.5 Br=6.092 Nc=2 Isc=0 IkT=0 Rc=1`  
`+Cjc=7.306p Mjc=.3416 Vjc=.75 Fc=.5 Cje=22.01p Mje=.377 Vje=.75`  
`+Tr=46.91n Tf=411.1p Itf=.6 Vtf=1.7 Xtf=3 Rb=10)`
    - Sono presenti anche sotto circuiti per la rappresentazione di componenti complessi come periferiche digitali e analogiche (opamp reali)

# Amplificatore invertente con ua741

Amplificatore invertente con opamp ua741

```
V1 1 0 AC 1 SIN(0 1m 1K)
```

```
R1 1 2 1K
```

```
R2 2 3 10K
```

```
R3 4 0 900
```

```
X1 4 2 5 6 3 UA741
```

```
Vp 5 0 15V
```

```
Vn 6 0 -15V
```


```
.LIB D:\Programmi\orcad_ST\Capture\Library\Pspice\eval.lib
```

```
.AC DEC 100 1 1MEG
```


```
.TRAN 1U 1M
```

```
.PROBE
```


```
.END
```



# Amplificatore invertente con ua741 -PROBE



# Amplificatore invertente con ua741 –AC PROBE



# Ricerca di un componente

- Per ricercare un componente bisogna aprire la libreria (sempre con pspice) e trovare il componente di interesse.
- Esempio del contenuto (eval.lib):

\* The following is a summary of parts in this library:

*	Part name	Part type
*	-----	-----
*	Q2N2222	NPN bipolar transistor
*	Q2N2907A	PNP bipolar transistor
*	Q2N3904	NPN bipolar transistor
*	Q2N3906	PNP bipolar transistor
*		
*	Q2N6052	PNP Darlington power transistor
*	Q2N6059	NPN Darlington power transistor

## Ricerca di un componente 2

- Nelle righe seguenti si può ancora trovare per esempio:

\* 7449 DECODER/DRIVER BCD-7 SEGMENT WITH OPEN-COLLECTOR OUTPUTS

\*

\* TTL LOGIC STANDARD TTL, S, LS DATA BOOK, APR 1988, TI

\* JLS 8-26-92 REMODELED USING LOGICEXP, PINDLY, AND CONSTRAINT DEVICES

```
.SUBCKT 7449 INA_I INB_I INC_I IND_I IBAR_I  
+ OUTA_O OUTB_O OUTC_O OUTD_O OUTF_O OUTG_O  
+ OPTIONAL: DPWR=$G_DPWR DGND=$G_DGND  
+ PARAMS: MNTYMXDLY=0 IO_LEVEL=0  
...continua
```

## Circuiti analogico-numeric

- Dal 1987 Pspice permette la simulazione di circuiti analogico-numeric e puramente numeric.
- Le analisi effettuabili sono tutte quelle già viste nel caso analogico
  - .DC
  - .AC
  - .TRAN

## Circuiti analogico-numeric 2

- I vari dispositivi numerici sono descritti da sottocircuiti e modelli. La loro chiamata si riduce alla quella di un sottocircuito.
  - Esempio:
    - X1 IN\_A IN\_B CLK CD4093B effettua la chiamata al componente CD4093B (CMOS schmitt trigger NAND gate).

## Circuiti analogico-numericì 3

- Pspice distingue 3 tipi di nodi:
  - Analogico
    - Se a questo sono connessi solo dispositivi analogici
  - Numerico
    - Se a questo sono connessi solo dispositivi numerici
  - Interfaccia
    - Se a questo sono connessi dispositivi analogici e numerici.
    - Pspice divide questo nodo in due nodi successivi, uno di tipo numerico-analogico ed uno di tipo analogico-numerico, inserendo apposite D/A A/D.

# Circuiti analogico-numeric 4

- I nodi che abbiamo utilizzato fino a questo punto si chiamano locali, in quanto hanno validità nella sola area del nostro file.cir
- Il nodo 0 invece è di tipo globale, in quanto è valido nel nostro file e in tutti quelli inclusi dal comando .LIB
- Un nodo globale deve cominciare con il simbolo: \$\_  
In aggiunta al nodo globale 0 Pspice offre anche altri tre nodi G:
  - \$D\_HI stato logico alto
  - \$D\_LO stato logico basso
  - \$D\_X stato logico non definito
- I dispositivi analogici NON possono essere connessi a questi tipo di nodi.
- **Tutti i dispositivi che prevedono un ingresso di abilitazione o di reset devono essere:**
  - **Disabilitati e riabilitati**
  - **Resettati e settati**
- Questo viene fatto attraverso un generatore di tensione ad hoc.

# X-OR

- Funzione logica:

$$Y = WX + W\bar{X}$$

EX-OR

X1 W 3 Y1 7408

X2 2 4 Y2 7408

X3 Y1 Y2 Y 7432

X4 W 2 7404

X5 X 3 7404

v1 W 0 5Vdc


v2 X 0 5vdc

.LIB D:\Programmi\orcad\_ST\Capture\Library\Pspice\eval.lib

.TRAN 1N 1U


.PROBE

.END




# X-OR - PROBE

X\$At0D  
W\$At0D  
Y  
Y1  
Y2




# Contatore binario 4 bit

```
Contatore binario
*.SUBCKT 7493A  CKA_I CKB_I R01_I R02_I
*+QA_O QB_O QC_O QD_O
X1 2 Q1 3 Q1 Q2 Q3 Q4 7493A
*Tensione di abilitazione
Ven 3 0 PWL(0 5 1U 5 1.1U 0)
*Clock del circuito
Vclk 2 0 PULSE(0 5 0 1N 1N 0.5U 1U)
.LIB D:\Programmi\orcad_ST\Capture\Library\Pspiceeval.lib
.TRAN 0.1U 18U
.PROBE
.END
```



# Contatore binario 4 bit - PROBE



# Generatore di gradini

Generatore di gradini (Convertitore D/A)

X1 2 Q1 3 3 Q1 Q2 Q3 Q4 7493A

\*Tensione di abilitazione

Ven 3 0 PWL(0 5 1U 5 1.1U 0)

\*Clock del circuito

Vclk 2 0 PULSE(0 5 0 1N 1N 0.5M 1M)

\*Carichi resistivi

R1 Q1 4 16K

R2 Q2 4 8K

R3 Q3 4 4K

R4 Q4 4 2K

X2 0 4 6 7 5 UA741

R5 4 5 1K

V+ 6 0 15V

V- 7 0 -15V


.LIB

D:\Programmi\orcad\_ST\Capture\Library\Pspice  
leva.lib

.TRAN 0.1M 18M

.PROBE

.END



# Generatore di gradini - PROBE

